
Revisando Conceitos:

Bit ‑ É a menor unidade de informação representável em computadores.

Byte ‑ Conjunto de 8 bits,  byte é a menor unidade de memória endereçável. Os bits dentro de um byte são numerados da direita para a esquerda começando com 0.

Word ‑ É a quantidade de bits que pode ser processada em qualquer operação de transferência de dados entre a CPU e a memória principal em um dado instante. Uma word é composta de 2 bytes no caso dos IBM‑PC.

Sistemas de Numeração:

Existem diversos sistemas de numeração, sendo mais conhecido o decimal ou de base 10. Como já é sabido, os computadores são máquinas binárias e portanto trabalham na base 2, onde os números são representados com apenas dois algarismos, o 0 e o 1. Por ser muito difícil trabalhar com números expressos na base 2, os seres humanos utilizam muitas vezes a base 16 (ou hexadecimal), pois é uma forma mais compacta para representar os números binários.

Existe uma relação direta de 4 dígitos binários para um dígito hexadecimal (24 = 16). Portanto, é possível representar 1 byte (8 bits) com dois algarismos hexadecimais.

Números  Hexadecimais:

São números expressos na base 16 e portanto dispõem de 16 algarismos para a sua representação. Os algarismos utilizados para representar os números hexadecimais e seus respectivos equivalentes decimais são: 0   1   2   3   4   5   6   7   8   9   A   B   C   D   E   F, onde A=10, B=11, C=12, D=13, E=14, F=15.

Para evitar confusões futuras, nós adotaremos a seguinte convenção para a representação de números:

‑ Números hexadecimais, serão terminados pela letra "h".    Ex: 10h.

‑ Números binários, terminarão com a letra "b".

- Números hexadecimais que comecem com letra deverão ser precedidos de 0. EX: 0B000h

Como Converter Números Hexadecimais em Decimais:

Os números decimais são formados da seguinte forma:


273 =
2 x  
100  = 
200

               
7 x  
10   =  
70
                    



3 x 
1    =   
3


            ‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑
                   



273             = 
273

  Analogamente:


D3h = 
D  =   
13 x 
16  = 
208

             
3  =    
3 x 
1   =   
3


          ‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑  



D3h = 


211

Convertendo Decimal Para Hexa:

Vejamos o que acontece quando dividimos um número decimal por 10:

             235 / 10 =  23   resto 5

              23 / 10  =  2   resto 3

               2 / 10  =  0   resto 2

Os dígitos de 235 aparecem como restos em ordem contrária. Portanto, para converter decimais para hexa, você deve dividir o número decimal por 16 e ir colhendo os restos da mesma maneira que foi feita para os números decimais.

    1069 / 16  = 66  resto 13

      66 / 16  =  4  resto  2

       4 / 16  =  0  resto  4

‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑        

       1069    =    42Dh

Notação Binária:

Como já foi dito, existe uma relação simples entre os números binários e os números hexadecimais: um dígito hexadecimal corresponde a quatro dígitos binários (24 = 16). A tabela abaixo permite converter qualquer número hexadecimal em binário e vice‑versa:

	Binário 
	Decimal 
	  Hexadecimal

	(0000) 
	0 
	0 

	(0001) 
	1 
	1 

	(0010) 
	2 
	2 

	(0011) 
	3 
	3 

	(0100) 
	4 
	4 

	(0101) 
	5 
	5 

	(0110) 
	6 
	6 

	(0111) 
	7 
	7 

	(1000) 
	8 
	8 

	(1001) 
	9 
	9 

	(1010) 
	10 
	A 

	(1011) 
	11 
	B 

	(1100) 
	12 
	C 

	(1101) 
	13 
	D 

	(1110) 
	14 
	E 

	(1111) 
	15 
	F 


Exemplo:  
0100
1101
0001
1100



4
D
1
C

Arquitetura Interna Do 8088:

· 14 registradores de 16 bits.

· Capaz de acessar 1Mb de memória.

· Capaz de acessar 65.535 portas de I/O

Registradores de Uso Geral:

AX

BX

CX

DX


(16 bits)
AH/AL
BH/BL
CH/CL
DH/DL

(8 bits)
AX: Acumulador


Usado em operações aritméticas.

BX: Base


Usado para indexar tabelas de memória (ex.: índice de vetores).

CX: Contador


Usado como contador de repetições em loop e movimentação repetitiva de dados.

DX: Dados


Uso geral.

Registradores de Segmento:

São usados para a identificação de segmentos específicos de memória.

CS: Code Segment


Localiza o segmento de código do programa que está sendo executado.


O valor do CS não é atualizável.

DS: Data Segment


Localiza o segmento de dados do programa.

SS: Stack Segment


Localiza o segmento de pilha do programa.

ES: Extra Segment


Usado para acessar um segmento de dados adicional.

Registradores de Deslocamento:

Usados em conjunto com registradores de segmento.

IP: Instruction Pointer

Usado em conjunto com o CS para apontar exatamente para o byte da próxima instrução a ser executada.

Não há acesso direto.

SP: Stack Pointer


Usado em conjunto com o SS para apontar o topo da pilha (stack).

BP: Base Pointer


Usado para indicar a base da pilha.


Particularmente importante para a passagem de parâmetros em linguagem de alto nível (como C).

SI & DI:  Source Index e Destination Index


Particularmente usados como instruções de string.


Podem ser usados para indexar tarefas de memória.

Registrador de Flags:

Consiste em um grupo individual de bits de controle (flag), os quais podem ser inspecionados um a um como dados comuns, além de poderem ser salvos ou recuperados como um grupo coordenado.

Possui 16 bits como os demais registradores.

OF: Overflow Flag


Setada quando ocorre overflow aritmético.

DF: Direction Flag


Setada para auto-incremento em instruções de string. 

IF: Interruption Flag


Permite que ocorram interrupções quando setada.


Pode ser setada pelo sistema ou pelo usuário.

TF: Trap Flag (debug)


Usada por debugadores para executar programas passo a passo.

SF: Signal Flag


Resetada (SF=0) quando um resultado for um número positivo ou zero e setada (SF=1) quando um resultado for negativo.

ZF: Zero Flag


Setada quando um resultado for igual a zero.

AF: Auxiliar Flag


Setada quando há “vai um” na metade inferior de um byte.

PF: Parity Flag


Setada quando o número de bits 1 de um resultado for par.

CF: Carry Flag


Setada se houver “vai um” no bit de maior ordem do resultado.


Também usada por instruções para tomadas de decisões.

Memória Segmentada:

· A memória é dividida em um número arbitrário de segmentos.

· O tamanho dos segmentos varia de 16 bytes a 64Kb.

· Cada segmento se inicia em uma posição perfeitamente divisível por 16.

· O endereço segmentado é sempre composto por duas words (32 bits).

· A anotação usada para representar um endereço é:  seg:desloc
· Exemplo: 


0041h : 0007h

(CS = 0041h  e  IP = 0007h)

endereço físico:

    00410h

+   0007h

======

 00417h

A Pilha:

· Toma conta do programa em execução.

· Pode servir para o armazenamento temporário de dados.

· Serve para a passagem de parâmetros.

· É usada de baixo (endereço mais alto) para cima (endereço mais baixo).

· Empilha e desempilha apenas words (16 bits).

· Controle quanto a “over flow” da pilha é responsabilidade do programador.

O Programa Hello

SEGMENT XE "SEGMENT"  e ENDS XE "ENDS" 
Sintaxe:

Nome SEGMENT
comandos e procedures...

Nome ENDS

· Marcam o início e o fim de um segmento, respectivamente.

· Um programa pode ter mais do que um segmento com o mesmo nome.

ASSUME XE "ASSUME" 
Sintaxe:

Assume Reg_Seg : Nome_Seg

· Associa um segmento a um registrador de segmento.

· Reg_Seg pode ser: CS, DS, ES, SS

PROC XE "PROC"  E ENDP XE "ENDP" 
Sintaxe:

Nome PROC [distância]

comandos...

Nome ENDP

· Marcam o início e o fim de uma procedure (função), respectivamente.

· O nome indicado para a procedure deve ser único em todo o programa.

· A distância pode ser NEAR (default) para uma procedure contida no segmento, ou FAR para uma procedure contida em outro segmento.

· A procedure deve conter ao menos um RET, que efetua o retorno da função.

LEA XE "LEA"  (Load Effective Address)

Sintaxe:

Lea destino, origem

Lógica:

destino ( endereço (origem)

Exemplo:

Lea Bx, var   (reg16, memória16)

MOV XE "MOV" 
Sintaxe:

Mov 
destino, origem

Lógica:

destino ( origem

Operações e Exemplos:

	Operação
	Exemplo

	registrador, 
registrador
	mov Bx, Cx

	memória, 
acumulador
	mov var, Al

	acumulador, 
memória
	mov Ax, var

	memória, 
registrador
	mov var, Si

	registrador, 
memória
	mov Si, var

	registrador, 
imediato
	mov var, 12

	reg_seg, 
reg16
	mov Ds, Ax

	reg16, 
reg_seg
	mov Ax, Ds

	memória, 
reg_seg
	mov var, Ds


INT XE "INT"  (Interrupt)

Sintaxe:

Int num_int

Lógica:

Empilha Flags

Limpa Trap Flag

Desabilita ints

Callfar int              (chama o procedimento da interrupção)

Observação:


A interrupção 21h contém os serviços do DOS.

A interrupção 10h contém os serviços da placa de vídeo.

No final da Int as flags são desempilhadas.

INT 21h, 09h XE "INT:INT 21h, 09h"   (Print String XE "Print String" )

Entrada:

Ah = 09h



Ds:Dx = Ponteiro para a string

Retorna:

Nada

Observação:

A string deve ser terminada por um ‘$’

INT 21h, 4Ch XE "INT:INT 21h, 4Ch"  (Terminate a Process XE "Terminate a Process" , Exit XE "Exit" )

Entrada:

Ah = 4Ch

 Al = Return code

Retorna:
 Nada

· Fecha todos os arquivos e retorna o controle ao chamador, liberando o espaço de memória usado para o DOS.

· É a maneira correta de se encerrar um programa em DOS (2.0 ou superior).

· O Return Code pode ser testado através da variável de sistema errorlevel.

END XE "END" 
· Indica o fim de um módulo.

Sintaxe:

End [expressão]

expressão ( indica o ponto de entrada do programa

DB e DW XE "DB / DW"  (Define Byte e Define Word)

· Aloca e inicializa um ou mais bytes de memória.

Sintaxe:

[nome] Db valor,valor2,...,...

nome ​​​​( cria uma variável quando especificado

valor ( pode ser:

	Tipo
	Exemplo

	Inteiro
	5

	String
	‘mensagem’

	Constante
	2*5

	Operador DUP
	10 DUP (?)

	Interrogação
	?


Observação:
Uma word não pode receber um byte e vice-versa.

DUP XE "DUP"  (Duplicate)

Sintaxe:

num Dup (valor,...,...)

Lógica:

Repete os valores especificados num vezes

Exemplo:

Db 200 Dup (1)
; 200 bytes inicializados com 1

Db 50 Dup (1,0)
; 100 bytes inicializados com 1 e 0

STACK XE "STACK"  (Segment Stack)

Sintaxe:

nome_seg Segment Stack
· Define um segmento de pilha.

· Os endereços neste segmento estão associados ao registrador SS.

Observações:

· Chamadas à uma interrupção usam pilha, portanto, atenção para a existência da pilha quando usar Int.

· O caracter ‘;’ indica ao compilador que a partir daí a linha é apenas comentário.

· As tabelas de exemplo sempre mostram todos os tipos aceitos pela instrução.

Hello.asm XE "Hello.asm" 
Code_Seg   Segment

           Assume CS:Code_Seg, DS:Data_Seg

Hello      Proc Far

           Mov  Ax, Data_Seg

           Mov  Ds, Ax

           Lea  Dx, Msg

           Mov  Ah,09

           Int  21h

           Mov  Ah,4Ch

           Int  21h

Hello      Endp

Code_Seg   Ends

Data_Seg   Segment

    Msg    Db 'Hello Assembly !$'

Data_Seg   Ends

Stack_Seg  SEGMENT STACK

   db      10 Dup('Pilha   ')

Stack_Seg  Ends

    End    Hello

O Programa Hello 2

CALL XE "CALL" 
· Chama uma procedure.

Sintaxe:

Call nome_procedure

Lógica:

Se Far  Call
{
Empilha CS


CS  segmento destino }

Empilha IP

IP  deslocamento destino

RET XE "RET"  (Return)

· Retorna de uma chamada de uma procedure.

· Comando obrigatório para retorno em Assembly.

Sintaxe:

Ret [valor]

Lógica:

Desempilha IP

Se RET Far

{
Desempilha CS }

[SP  SP + valor]

PUSH XE "PUSH" 
· Empilha uma word

Sintaxe:   
Push origem
Lógica:

SP  SP ( 2

(SP)  origem
	Operações
	Exemplo

	Reg
	PUSH BX

	Memória
	PUSH Var

	Reg_Seg (exceto IP)
	PUSH ES


POP XE "POP" 
· Desempilha uma word

Sintaxe:   
Pop destino
Lógica:

destino  (SP)

SP  SP + 2

	Operações
	Exemplo

	Reg
	POP SI

	Memória
	POP Var

	Reg_Seg 

(exceto CS e IP)
	POP DS


XOR XE "XOR"  (Ou Exclusivo)

Sintaxe:

Xor destino, origem

Lógica:

destino ( destino xor origem (bit a bit)

Flags:

O D I T S Z A P C

0       * * * * 0

Tabela Verdade:

	Destino
	Origem
	Resultado

	0
	0
	0

	0
	1
	1

	1
	0
	1

	1
	1
	0



Operações e Exemplos:

	Operação
	Exemplo

	registrador, 
registrador
	Xor Bx, Cx

	registrador, 
memória
	Xor Cl, var

	memória, 
registrador
	Xor var, Dx

	acumulador, 
imediato
	Xor Al, 10101111b

	registrador, 
imediato
	Xor Si, 0CAh

	memória, 
imediato
	Xor var, 3


Observação:

· Pode ser usado para zerar um registrador ou uma variável (exemplo: Xor Cx, Cx), tendo o mesmo efeito que mover zero, porém, com menos ciclos de máquina.

SUB XE "SUB"  (Subtract)

Sintaxe:
Sub destino, origem
Lógica:
destino ( destino ( origem
Flags:
O D I T S Z A P C

*       * * * * *
Operações e Exemplos:

	Operação
	Exemplo

	registrador, 
registrador
	Sub Dx, Bx

	registrador, 
memória
	Sub Bx, var

	memória,
registrador
	Sub var, Si

	acumulador,
imediato
	Sub Al, 5

	registrador,
imediato
	Sub Dx, 3

	memória, 
imediato
	Sub var, 0Fh


ADD XE "ADD" 
Sintaxe:
Add destino, origem
Lógica:
destino ( destino + origem
Flags:
O D I T S Z A P C

*       * * * * *
Operações e Exemplos:

	Operação
	Exemplo

	registrador, 
registrador
	Add Dx, Bx

	registrador, 
memória
	Add Bx, var

	memória,
registrador
	Add var, Si

	acumulador,
imediato
	Add Al, 5

	registrador,
imediato
	Add Dx, 3

	memória, 
imediato
	Add var, 0Fh


SHR XE "SHR"  (Shift Right)

Sintaxe:
Shr destino, quantidade
· Desloca os bits do destino para a direita, o registrador CF recebe o bit de menor ordem e o bit de maior ordem recebe valor 0. 

Lógica:
0 ( destino ( CF

Flags:
O D I T S Z A P C

*       * * ? * *
Operações e Exemplos:

	Operação
	Exemplo

	registrador, 
1
	SHR  SI, 1

	registrador,  
CL
	SHR  BX, CL

	memória,  
1
	SHR  Var, 1

	memória,  
CL
	SHR  Var, CL


SHL XE "SHL"  (Shift Left)

Sintaxe:
Shl destino, quantidade
· Desloca os bits do destino para a esquerda, o registrador CF recebe o bit de maior ordem e o bit de menor ordem recebe valor 0. 

Lógica:
CF ( destino ( 0

Flags:
O D I T S Z A P C

*       * * ? * *
Operações e Exemplos:

	Operação
	Exemplo

	registrador, 
1
	SHL  SI, 1

	registrador,  
CL
	SHL  BX, CL

	memória,  
1
	SHL  Var, 1

	memória,  
CL
	SHL  Var, CL


Observações:

· O único valor imediato que pode ser usado é 1. Para deslocar mais de um bit de uma única vez é necessário usar o registrador Cl com o total de bits a serem deslocados.

· Usados para multiplicação e divisão por potências de 2.

· Em ambos a OF (overflow flag) é setada se houver uma troca de sinal, ou seja, se o bit de mais alta ordem trocou pois números com sinal são guardados em complemento a 2.

· No caso de rotação de mais de um bit, CF fica com o valor do último rotacionado.

INT 10h, 06h XE "INT:INT 10h, 06h"  (Scroll Window Up XE "Scroll Window Up" )

Entrada:

Ah
06h

Al
Número de linhas a rolar ou


0 para limpar a região

Bh
Atributo para novas linhas

Ch
Linha superior

Cl
Coluna esquerda

Dh
Linha inferior

Dl
Coluna direita

Retorna:
nada

Observações:

· Int 10h, 07h (Scroll Window Down) tem as mesmas entradas.

· Pode ser usada para limpar toda a tela.

INT 10h, 0Fh XE "INT:INT 10h, 0Fh"  (Get Current Video Mode XE "Get Current Video Mode" )

Entrada: 
Ah
0Fh
Retorna:
Al
Modo de vídeo


Ah
Número de caracteres por linha


Bh
Página de vídeo ativa

INT 10h, 02h XE "INT:INT 10h, 02h"  (Set Cursor Position XE "Set Cursor Position" )

Entrada:

Ah
02h

Bh
Página de vídeo

Dh
linha

Dl
coluna

Retorna:
nada

INT 21h, 08h XE "INT:INT 21h, 08h"  (Console Input Without Echo XE "Console Input Without Echo" )

Entrada:

Ah
08h

Retorna:

Al
char

Observação:

· TECLA ESPECIAL: Se a tecla for especial (exemplo: F1) Al retornará 0. Para pegar o código da tecla especial tem que ser chamada uma nova int 21h, 08h.

Observações:

· O número dos serviços de interrupção deverá estar sempre no Ah.

· Usar Xor para zerar, Shr para dividir por 2 e Shl para multiplicar por 2 são maneiras de se programar economizando ciclos de máquina.

· Se você pretende fazer com que sua rotina salve no início os registradores usados e os restaure no final, lembre-se de que existem registradores que são usados como retorno de valores de uma interrupção.

· A rotina Encontra_Pos do programa Hello2 pode ser otimizada!

· Push é mais rápido que salvar usando Mov.

· Int 10h é a interrupção de vídeo (pertence ao grupo de interrupções da Bios).

Hello2.asm XE "Hello2.asm" 
Code_Seg   Segment

           Assume Cs:Code_Seg,

                  Ds:Data_Seg

Hello2     Proc   Near

           Mov    Ax, Data_Seg

           Mov    Ds, Ax

           Call   Limpa_Tela

           Call   Encontra_Pos

           Call   Say_Hello

           Call   Espera_Tecla

           Call   Limpa_Tela

           Mov    Ah,4Ch

           Mov    Al,00

           Int    21h

Hello2     Endp

Limpa_Tela Proc   Near

           Push   Ax

           Push   Bx

           Push   Cx

           Push   Dx

           Mov    Ax,0600h

           Mov    Bh,07

           Xor    Cx,Cx

           Mov    Dh,24

           Mov    Dl,79

           Int    10h

           Pop    Dx

           Pop    Cx

           Pop    Bx

           Pop    Ax

           Ret

Limpa_Tela EndP

Encontra_Pos  Proc  Near

              Push  Ax

              Push  Bx

              Mov   Ah,0Fh

              Int   10h

              Mov   X,Ah

              Sub   X,17

              Shr   X,1

              Pop   Bx

              Pop   Ax

              Ret

Encontra_Pos  Endp

Say_hello  Proc   Near

           Push   Ax

           Push   Dx

           Call   SetCursor

           Mov    Ah,09

           Lea    Dx,Msg

           Int    21h

           Pop    Dx

           Pop    Ax

           Ret

Say_hello  EndP

SetCursor  Proc   Near

           Push   Ax

           Push   Bx

           Push   Dx

           Xor    Bh,Bh

           Mov    Ah,02

           Mov    Dh,Y

           Mov    Dl,X

           Int    10h

           Pop    Dx

           Pop    Bx

           Pop    Ax

           Ret

SetCursor  Endp

Espera_Tecla Proc  Near

             Push  Ax

             Mov   Ah,08

             Int   21h

             Pop   Ax

             Ret

Espera_Tecla Endp

Code_Seg   Ends

Data_Seg   Segment

  Msg Db 'Hello Assembler !$'

  X   Db ?

  Y   Db 12

Data_Seg   Ends

Stack_Seg  SEGMENT STACK

  db       10 Dup(' ')

Stack_Seg  Ends

    End    Hello2

O PROGRAMA TECLA

Rótulo XE "Rótulo" :

Sintaxe:
nome_rotulo:

· Diretiva de compilação que marca uma posição dentro do código.

· Toda vez que o compilador encontrar o nome do rótulo ele irá substituir pelo endereço onde se encontra o rótulo (deslocamento).

CMP XE "CMP"  (Compare)

Sintaxe:
Cmp destino, origem
Lógica:


Flags setadas de acordo com o resultado de: destino ( origem
Flags:
O D I T S Z A P C

*       * * * * *

Operações e Exemplos:

	Operação
	Exemplo

	Registrador, 
registrador
	Cmp Cx, Bx

	Registrador,
imediato
	Cmp Si, 3

	Acumulador,
imediato
	Cmp Al, 0Fh

	Registrador, 
memória
	Cmp Bx, var

	Memória,
registrador
	Cmp var, Cx

	Memória, 
imediato
	Cmp var, 8


JB XE "JB, JC, JNAE"  (Jump if Below)

Sintaxe:
Jb rótulo
Lógica:

Se CF (carry flag) = 1


Salta para rótulo
Observação:

JB = JC (Jump if Carry) = JNAE (Jump if Not Above or Equal)


JA XE "JA, JNBE"  (Jump if Above)

Sintaxe:
Ja rótulo
Lógica:

Se CF (carry flag) = 0 && ZF (zero flag) = 0


Salta para rótulo
Observação:

JA = JNBE (Jump if Not Below or Equal)

JMP XE "JMP"  (Jump)

· Salto incondicional

Sintaxe:
Jmp rótulo
Lógica:
Salta para rótulo
Observação:

· O Salto pode ser longo, inclusive para outros segmentos.

INC XE "INC"  (Increment)

Sintaxe:
Inc destino
Lógica:
destino ( destino + 1

Flags:
O D I T S Z A P C

*       * * * *

Operações e Exemplos:

	Operação
	Exemplo

	registrador 16 bits
	Inc Bx

	Registrador 8 bits
	Inc Cl

	Memória
	Inc var


Observação:

· Para decrementar use o comando DEC XE "DEC" , cuja sintaxe é a mesma de INC.

LOOP XE "LOOP" 
Sintaxe:
Loop rótulo
Lógica:


Cx ( Cx ( 1


Se ( Cx != 0 )



Jmp rótulo 

Observação:

· Se Cx = 0 no início serão feitos 65.536 loops.


LABEL XE "LABEL" 
· Cria um rótulo ou var

Sintaxe:
nome label tipo
Observação:

· tipo pode ser:

· byte, word, dword, etc.

· near, far

Exemplo:

Byte_buffer label byte

Word_buffer dw 512 dup (?)

INT 21h, 0Ah XE "INT:INT 21h, 0Ah"  (Buffer Input XE "Buffer Input" )

· Lê os caracteres, ecoando-os na tela se o tamanho máximo ainda não tiver estourado, até que <Enter> seja pressionado.

Entrada:

Ah
0Ah

Ds:Dx
( ponteiro para buffer

offset 0 do buffer ( tamanho máximo do buffer + 1

Retorna:

offset 1 do buffer ( número de caracteres efetivamente lidos (sem contar o <Enter>)

offset 2 do buffer em diante ( caracteres lidos incluindo o CR (Carriage Return = <Enter>)
Observação:

· Não lê teclas de função.

· Edição rudimentar.

· Dados só podem ser criticados ao final.

· Quando o limite de caracteres a serem lidos estoura, o beep toca.

INT 21h, 02h XE "INT:INT 21h, 02h"  (Char Output XE "Char Output" )

· Imprime o caracter ou o executa, se for do tipo beep, line feed ou assemelhados.

Entrada:

Ah
02h

Dl
caracter

Observações:

· Os “jumps” condicionais (Jb, Jae, etc.) têm um limite de 127 bytes de distância até o rótulo. O Jmp não tem limite (pode ser inclusive em outro segmento), portanto para dar um salto condicional maior que 127 bytes usa-se comumente um recurso denominado trampolim, que é o uso de um “jump” condicional apontando para um Jmp (incondicional) para que a distância do rótulo seja vencida.

· Offset = deslocamento. Deslocamento também pode ser representado em assembly pela presença de colchetes. Exemplo: var[3]

· Ja e Jb são usados para números positivos; para números com sinal use JG (Jump if Greater) e JL (Jump if Less).

· O comando “if” de C em Assembly é a junção do CMP com um “jump” condicional.

· O camando “for” de C em Assembly é o comando Loop que repete um bloco Cx vezes.

· Ao usar o Loop devemos estar atentos quanto a alterações em Cx para não atrapalhar o controle de repetição.

· Os jumps condicionais devem estar imediatamente após o Cmp, porque outras instruções podem alterar o resultado das flags.

· Os únicos registradores que podem ser usados como indexadores de memória são: Bx, Si e Di. Exemplo: var[Bx]

· O assembly permite que você faça jumps para uma procedure, no entanto não se deve fazer uso deste recurso por dois motivos: 1) O Ret não irá funcionar, pois ele espera encontrar na pilha o endereço de retorno que é empilhado pelo Call, os jumps não empilham endereço. 2) Mesmo que você resolva o problema anterior, ao fazer jumps para procedures você estará perdendo a estruturação do seu programa e portanto sua qualidade. Para chamar uma procedure use apenas o comando Call.

Tecla.asm XE "Tecla.asm" 
Code_Seg   Segment

           Assume Cs:Code_Seg,

                  Ds:Data_Seg

Tecla   Proc   Near

        Mov    Ax, Data_Seg

        Mov    Ds, Ax

        Call   Le_Texto

        Call   Conv_Maiuscula

        Mov    Ah,4Ch

        Int    21h

Tecla   Endp

Le_Texto   Proc   Near

           Push   Ax

           Push   DX

           Lea    Dx, Texto

           Mov    Ah,0Ah

           Int    21h

           Pop    Dx

           Pop    Ax

           Ret

Le_Texto   Endp

Conv_Maiuscula  Proc   Near

         Push   Ax

         Push   Bx

         Push   Cx

         Xor    Ch,Ch

         Mov    Cl,Car_Lidos

         Lea    Bx,String

Loop_Conversao:

         Mov    Al,[Bx]

         Cmp    Al,'a'

         Jb     Prox_Byte

         Cmp    Al,'z'

         Ja     Prox_Byte

         Sub    Al,20h

         Mov    [Bx],Al


Prox_Byte:

         Inc    Bx

         Loop   Loop_Conversao

         Pop    Cx

         Pop    Bx

         Pop    Ax

         Ret

Conv_Maiuscula    EndP

Code_Seg   Ends

Data_Seg   Segment

Texto  Label Byte

    Tam_String  Db 31

    Car_Lidos   Db 0

    String      Db 31 Dup(' ')

Data_Seg   Ends

Stack_Seg  SEGMENT STACK

   db      10 Dup('Pilha   ')

Stack_Seg  Ends

    End    Tecla

O PROGRAMA IMPBIN

JE XE "JE, JZ"  (Jump if Equal)

Sintaxe:
Je rótulo
Lógica:

Se ZF (Zero flag) = 1


Salta para rótulo

Observações:

· JE = JZ (Jump if Zero).

·  JNE XE "JNE"  (Jump if Not Equal) teste para o salto é inverso do JE.
OR XE "OR"  

· Ou lógico

Sintaxe:
Or destino, origem
Lógica:
destino ( destino or origem

(bit a bit)

Flags:
O D I T S Z A P C
0 * * ? * 0

Tabela Verdade:

	Destino
	Origem
	Resultado

	0
	0
	0

	0
	1
	1

	1
	0
	1

	1
	1
	1



Operações e Exemplos: 

Vide Xor na página 15.

Observações:

· Por gastar menos ciclos de máquina que o Cmp, é usado para comparar com 0: Or x,x

· Existe também a operação lógica AND XE "AND" , cuja tabela verdade é:

	Destino
	Origem
	Resultado

	0
	0
	0

	0
	1
	0

	1
	0
	0

	1
	1
	1


· Existe ainda a operação NOT XE "NOT" , cuja sintaxe só tem destino (sem origem).

ADC XE "ADC"  (Add With Carry)

Sintaxe:
Adc destino, origem
Lógica:
destino ( destino + origem + CF

CLC XE "CLC, CLD, CLI"  (Clear Carry Flag)

Sintaxe:
CLC
Lógica:
CF ( 0

Flags:
O D I T S Z A P C

                 0
Observação:

· As flags DF e IF também podem ser zeradas através das instruções: CLD e CLI, respectivamente.

(Ex: CLD ( Clear Direction Flag)


STC XE "STC, STD, STI"  (Set Carry Flag)

Sintaxe:
STC
Lógica:
CF ( 1

Flags:
O D I T S Z A P C

                 1
Observação:

· As flags DF e IF também podem ser setadas (receberem valor 1) através das instruções: STD e STI, respectivamente.

(Ex: STI ( Set Interruption Flag)

RCL XE "RCL"  (Rotate Through Carry Left)

Sintaxe:
RCL destino, quantidade
Lógica:
CF ( destino ( 

Flags:
O D I T S Z A P C

*               *
Operações e Exemplos:

	Operação
	Exemplo

	registrador, 1
	RCL AX, 1

	registrador, CL
	RCL  BX, CL

	memória, 1
	RCL  Var, 1

	memória, CL
	RCL  Var, CL


Observação:

· Assim como existem o SHR e o SHL, também existe o RCR XE "RCR"  (Rotate Through Carry Right).

Observação:

· O comando Push não pode ser usado para empilhar vários registradores ao mesmo tempo. Apenas por uma questão de melhor visualização e aproveitamento do espaço usaremos em algumas das listagens dos programas: Push Ax, Bx, Cx.

Impbin.asm XE "Impbin.asm"  

Code_Seg   Segment

    Assume Cs:Code_Seg,

           Ds:Data_Seg

Hexa_Bin   Proc Near

    Mov    Ax, Data_Seg

    Mov    Ds, Ax

Loop_Texto:

    Call   Le_Texto

    Call   Conv_Maiuscula

    Call   Valida_Texto

    Jc     Loop_Texto

    Call   Imp_Bin

    Mov    Ah,4Ch

    Int    21h

Hexa_Bin   Endp

Le_Texto   Proc Near

    Push   Ax,Dx

    Lea    Dx, Texto

Loop_Leitura:

    Mov    Ah,0Ah

    Int    21h

    Cmp    Car_Lidos,0

    Je     Loop_Leitura

    Pop    Dx,Ax

    Ret

Le_Texto   Endp

Conv_Maiuscula Proc Near

;idem tecla.asm

Conv_Maiuscula EndP

Valida_Texto   Proc Near

    Push   Ax,Cx

    Mov    Al,String[0]

    Cmp    Al,'0'

    Jb     Erro_Digito

    Cmp    Al,'9'

    Ja     Tenta_Letra

    Sub    Al,'0'

    Mov    Ah,Al

    Jmp    Segundo_Digito

Tenta_Letra:

    Cmp    Al,'A'

    Jb     Erro_Digito

    Cmp    Al,'F'

    Ja     Erro_Digito

    Sub    Al,'A'-10

    Mov    Ah,Al

Segundo_Digito:

    Cmp    Car_Lidos,2

    Jb     Fim_Digitos

    Mov    Al,String[1]

    Cmp    Al,'0'

    Jb     Erro_Digito

    Cmp    Al,'9'

    Ja     Tenta_Letra2

    Sub    Al,'0'

    Jmp    Acerta_Num

Tenta_Letra2:

    Cmp    Al,'A'

    Jb     Erro_Digito

    Cmp    Al,'F'

    Ja     Erro_Digito

    Sub    Al,'A'-10

Acerta_Num:

    Mov    Cx,4

    Shl    Ah,Cl

    Or     Ah,Al

    Mov    Dig_Hexa,Ah

    Clc

    Jmp    Fim_Valida

Erro_Digito:

    Stc

    Jmp    Fim_Valida

Fim_Digitos:

    Clc

    Mov    Dig_Hexa,Al

Fim_Valida:

    Pop    Cx,Ax

    Ret

Valida_Texto      Endp

Imp_Bin    Proc   Near

    Push   Ax,Bx,Cx,Dx

    Mov    Ah,02

    Mov    Bl,Dig_Hexa

    Mov    Cx,8

Loop_ImpBin:

    Mov    Dl,0

    Rcl    Bl,1

    Adc    Dl,'0'

    Int    21h

    Loop   Loop_ImpBin

    Pop    Dx,Cx,Bx,Ax

    Ret

Imp_Bin    Endp

Code_Seg   Ends

Data_Seg   Segment

    Texto  Label Byte

    Tam_String  Db 3

    Car_Lidos   Db 0

    String      Db 3 Dup(' ')

    Dig_Hexa    Db ?

Data_Seg   Ends

Stack_Seg  SEGMENT STACK

   db  10 Dup('Pilha   ')

Stack_Seg  Ends

End    Hexa_Bin

Mul, Div e Exercícios

MUL XE "MUL"  (Multiply)

· Multiplicação sem sinal

Sintaxe:
Mul origem
Lógica:

Se (origem == byte)

AX = origem * AL 

Se (origem == word)

DX:AX = origem  * AX

Flags:
O D I T S Z A P C

*       ? ? ? ? *

Operações e Exemplos:

	Operando
	Exemplo

	imediato
	MUL 20

	   reg8
	MUL CH

	   reg16
	MUL BX

	   mem8
	MUL VarByte

	   mem16


	MUL VarWord



DIV XE “DIV”  (Divide) 

· Divisão sem sinal

Sintaxe:
Div origem
Lógica:

Se (origem == byte)


AL =  AX / origem

AH  = resto

Se (origem == 
ord)


AX  =  DX:AX / origem 


DX  =  resto

Flags:
O D I T S Z A P C

?       ? ? ? ? ?

Operações e Exemplos:

	Operando
	Exemplo

	imediato
	DIV 16

	   reg8
	DIV BL

	   reg16
	DIV BX

	   mem8
	DIV VarByte

	   mem16
	DIV VarWord


Observações:

· Para números com sinal use as instruções IMUL e IDIV.

· Se o resultado da divisão não couber é gerada a interrupção de divisão por 0. 

Exercícios:

1)  Faça uma rotina em Assembly que troque o valor do registrador AH com o valor do registrador BH utilizando apenas instruções lógicas: AND, OR ou XOR. Você pode utilizar outros registradores para guardar valores intermediários.

2) Faça uma rotina em Assembly que retorne em AX o maior valor de uma região na memória. A região deverá ser referenciada pela variável Tabela (tipo byte) que terá um tamanho fixo de 256 bytes. Tabela estará definida da seguinte forma:




Tabela db 0,4,0,7,255,4,3,..........., 200.
3) Faça um programa completo em Assembly que leia 5 algarismos. Para cada algarismo lido deve ser mostrado o caracter * ao invés do próprio algarismo. Caso o usuário digite algo diferente de algarismos o programa deve emitir um “bell”. Ao final da digitação o programa mostrará então quais foram os algarismos inseridos.

